

EBNF vir TOYJ

RW144

2005

EBNF

```
Program      = "class" id "{" { VarDecl } { Command } "}" .
VarDecl      = Type id { "," id } ";" .
Type         = "int" | "boolean" .
Block        = "{" { Command } "}" .
Command      = SimpleCommand | CompoundCommand | Block .
SimpleCommand = Out ";" | Assignment ";" | ";" .
CompoundCommand = If | While | For .
Assignment   = id "=" Expr | id "[" Expr "]" "=" Expr .
If           = "if" "(" Expr ")" Command
              [ "else" Command ] .
While        = "while" "(" Expr ")" Command .
For          = "for" "(" id "=" Expr ";" Expr ")" Command .
Out          = "print" "(" Expr ")" .
Expr         = Simple [ Relation Simple ] .
Relation     = "==" | "!=" | "<" | ">" | "<=" | ">=" .
Simple       = [ "+" | "-" ] Term { AddOp Term } .
AddOp        = "+" | "-" | "||" .
Term         = Factor { MulOp Factor } .
MulOp        = "*" | "/" | "&&" .
Factor       = number | id [ "[" Expr "]" ] | "(" Expr ")" |
              "!" Factor | "true" | "false" .
id           = letter { letter | digit } .
number       = digit { digit } .
letter       = "a" ... "z" | "A" ... "Z" .
digit        = "0" ... "9" .
```

Notas

- Kommentaar moet deur die leksikale ontleder herken en verwyder word. Twee tipes kommentaarstellings kom voor. Die eerste tipe begin met "//" en eindig aan die einde van die lyn waarop dit voorkom. Die tweede tipe begin met "/*" en eindig met "*/". Laasgenoemde tipe kommentaar kan ook genes word.

Voorbeelde van ToyJ programme

- `class c1 { }`
- `class c2
{ }`
- `class c3 {
 print (222);
}`
- `class c4 {
 id1 = 222;
}`
- `class c5 {
 while (222) { } ;
}`
- `// Program wat 1 2 3 4 5 vertoon m.b.v. for
class count1 {
 int number;
 for (number = 1 ; 5)
 print (number);
}`
- `// Program wat 1 2 3 4 5 vertoon m.b.v. while
class count2 {
 int number;
 number = 1;
 while (number < 6) {
 print (number);
 number = number + 1;
 }
}`